



Regex Module

smlhelp.github.io – Auxiliary Library

Jacob Neumann

July 2021

Contents

1	Preliminary Declarations & Main Definitions	3
2	Matching Specification	5
3	Matching Implementation	6
4	Zero, One, Const Correctness	8
5	Plus Correctness	9
6	Times Correctness	10
7	Star Correctness	11
8	match and LL Correctness	12
9	Reduction, Representation, and Printing Specification	12

Note 1

Throughout, `Sigma` is assumed to be some equality type.

Note 2

This document assumes the definitions and declarations of <https://github.com/smlhelp/aux-library/blob/main/documentation/language.pdf>, and assumes that the `Language` structure has been opened.

In particular, we have

Defn. 3

aux-library: Language.sml

```
31 type 'S language = 'S list -> bool
```

A total function `D : Sigma list -> bool` is said to **compute** a set L of values of type `Sigma list` if

$$D\ cs \implies \text{true} \quad \text{iff} \quad cs \in L$$

i.e. we can run `D` on any value `cs : Sigma list` and get back either `true` or `false`, which correctly indicates that either $cs \in L$ or $cs \notin L$, respectively.

1 Preliminary Declarations & Main Definitions

Declaration 4

aux-library: Regexp.sml

```

29  datatype ''S regexp =
30      Zero
31      | One
32      | Const of ''S
33      | Plus of ''S regexp * ''S regexp
34      | Times of ''S regexp * ''S regexp
35      | Star of ''S regexp

```

Defn. 5 (Language)

Given a value $R : \text{Sigma } \text{regexp}$, define the **language of R** to be the set of values $\mathcal{L}(R) \subseteq \text{Values}(\text{Sigma } \text{list})$ given recursively by:

$$\begin{aligned}
 \mathcal{L}(\text{Zero}) &= \emptyset \\
 \mathcal{L}(\text{One}) &= \{[]\} \\
 \mathcal{L}(\text{Const}(c)) &= \{[c]\} \\
 \mathcal{L}(\text{Plus}(r_1, r_2)) &= \mathcal{L}(r_1) \cup \mathcal{L}(r_2) \\
 \mathcal{L}(\text{Times}(r_1, r_2)) &= \{v_1 @ v_2 \mid v_1 \in \mathcal{L}(r_1) \text{ and } v_2 \in \mathcal{L}(r_2)\} \\
 \mathcal{L}(\text{Star}(r)) &= \{v_1 @ v_2 @ \dots @ v_n \mid n \in \mathbb{N}, v_1, v_2, \dots, v_n \in \mathcal{L}(r)\}
 \end{aligned}$$

Defn. 6

A total function $D : \text{Sigma } \text{regexp} \rightarrow \text{Sigma } \text{language}$ is said to **compute the regular expression decision problem** (over the alphabet Sigma) if

$$D \ R \ cs \implies \text{true} \quad \text{if and only if} \quad cs \in \mathcal{L}(R)$$

for all $R : \text{Sigma } \text{regexp}$ and all $cs : \text{Sigma } \text{list}$.

Defn. 7

A pair $(p, s) : \text{Sigma } \text{list} * \text{Sigma } \text{list}$ is said to be a **splitting** of $cs : \text{Sigma } \text{list}$ if

$$cs \cong p @ s.$$

In such a splitting, p is called the **prefix** and s the **suffix**.

Declaration 8

aux-library: Regexp.sml

47

`exception NoMatch`**Defn. 9**

For any type t , we'll say that a function

$$k : \text{Sigma list} * \text{Sigma list} \rightarrow t$$

is **almost total** if, for all $(p, s) : \text{Sigma list} * \text{Sigma list}$, either

$$k(p, s) \hookrightarrow v \text{ for some value } v \quad \text{or} \quad k(p, s) \text{ raises NoMatch.}$$

If $k(p, s) \hookrightarrow v$, then we say that **k accepts (p, s) with result v** .

2 Matching Specification

Declaration 10 (Depth)

Define the **depth** of $R : \text{Sigma regexp}$ recursively by

aux-library: Regexp.sml

```

37 fun depth Zero = 0
38   | depth One = 0
39   | depth (Const(_)) = 0
40   | depth (Plus(R1,R2)) =
41       1 + Int.max(depth R1,depth R2)
42   | depth (Times(R1,R2)) =
43       1 + Int.max(depth R1,depth R2)
44   | depth (Star(R)) =
45       1 + depth R

```

Value Spec

```

match : ''a regexp
      -> ''a list
      -> (''a list * ''a list -> 'b)
      -> 'b

```

REQUIRES: k is almost total (Defn. 9)

ENSURES:

$$\text{match } r \text{ } cs \text{ } k \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs \text{ such} \\ & \text{that } p \in \mathcal{L}(r) \text{ and } k \text{ accepts } (p, s) \\ & \text{with result } v. \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

Value Spec

$LL : ''S \text{ regexp} \rightarrow ''S \text{ language}$

ENSURES: LL computes the regular expression decision problem: for all $R : \text{Sigma regexp}$, $(LL \ R) : \text{Sigma list} \rightarrow \text{bool}$ is a total function such that

$$LL \ R \ cs \implies \text{true} \quad \text{iff} \quad cs \in \mathcal{L}(R)$$

3 Matching Implementation

aux-library: Regexp.sml

```
29 datatype ''S regexp =
30     Zero
31     | One
32     | Const of ''S
33     | Plus of ''S regexp * ''S regexp
34     | Times of ''S regexp * ''S regexp
35     | Star of ''S regexp
36
37 fun depth Zero = 0
38     | depth One = 0
39     | depth (Const(_)) = 0
40     | depth (Plus(R1,R2)) =
41         1 + Int.max(depth R1,depth R2)
42     | depth (Times(R1,R2)) =
43         1 + Int.max(depth R1,depth R2)
44     | depth (Star(R)) =
45         1 + depth R
46
47 exception NoMatch
```

aux-library: Regexp.sml

```

49 fun match Zero _ _ = raise NoMatch
50   | match One cs k = k([],cs)
51   | match (Const(c)) [] k = raise NoMatch
52   | match (Const(c)) (c'::cs') k =
53       if c=c'
54       then k([c'], cs')
55       else raise NoMatch
56   | match (Plus(R1,R2)) cs k =
57       (match R1 cs k
58        handle NoMatch => match R2 cs k)
59   | match (Times(R1,R2)) cs k =
60       match R1 cs (fn (res',cs') =>
61         match R2 cs' (fn (res'',cs'') =>
62           k(res'@res'',cs'')))
63   | match (Star(R)) cs k =
64       k([],cs)
65       handle NoMatch =>
66         match R cs (fn (res',cs') =>
67           if (cs = cs')
68           then raise NoMatch
69           else
70             match (Star(R)) cs' (fn (res'',cs'') =>
71               k(res'@res'',cs'')))
72
73 val LL = fn R => fn s =>
74   match R s (fn (_,[]) => true | _ => raise NoMatch)
75   handle NoMatch => false

```

4 Zero, One, Const Correctness

Lemma 11

For any almost total k and any $cs : \text{Sigma list}$,

$$\text{match Zero } cs \ k \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs \text{ such} \\ & \text{that } p \in \mathcal{L}(\text{Zero}) \text{ and } k \text{ accepts } (p, s) \text{ with result } v. \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

Proof. Since $\mathcal{L}(\text{Zero}) = \emptyset$, there cannot be any splitting of any cs whose prefix is in $\mathcal{L}(\text{Zero})$. Therefore, this spec says that `match Zero cs k` always ought to raise `NoMatch`, which it does. **Spec Satisfied!**

Cor. 12

For any almost total k , `(fn cs => match Zero cs k)` is almost total.

Lemma 13

For any almost total k and any $cs : \text{Sigma list}$,

$$\text{match One } cs \ k \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs \text{ such} \\ & \text{that } p \in \mathcal{L}(\text{One}) \text{ and } k \text{ accepts } (p, s) \text{ with result } v. \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

Proof. Since $\mathcal{L}(\text{One}) = \{ [] \}$, the only splitting of cs which we need to consider is $([], cs)$. But notice that

$$k([], cs) \cong \begin{cases} v & \text{if } k \text{ accepts } ([], cs) \text{ with result } v \\ \text{raise NoMatch} & \text{if } k \text{ rejects } ([], cs) \end{cases}$$

so we have **Spec Satisfied!** by putting `match One cs k = k([], cs)`. \square

Cor. 14

For any almost total k , `(fn cs => match One cs k)` is almost total.

Lemma 15

For any almost total k , any $cs : \text{Sigma list}$, and any value $c : \text{Sigma}$

$$\text{match } (\text{Const } c) \text{ } cs \text{ } k \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs \text{ such that } p \in \mathcal{L}(\text{Const } c) \text{ and } k \text{ accepts } (p, s) \text{ with result } v. \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

Proof. Begin by recalling that $\mathcal{L}(\text{Const } c) = \{[c]\}$. Break into two cases: $cs = []$ and $cs = c' :: cs'$.

If $cs = []$, then there is no prefix of cs in $\mathcal{L}(\text{Const } c)$, so $\text{match } (\text{Const } c) \text{ } [] \text{ } k$ should raise `NoMatch`. **Spec Satisfied!**

If $cs = c' :: cs'$, then break into two further cases: either $c = c'$ or $c \neq c'$. In the latter case, there is no prefix of cs in $\mathcal{L}(\text{Const } c)$, so again `NoMatch` should be raised.

Spec Satisfied! However, if $c = c'$, then $[c'] \in \mathcal{L}(\text{Const } c)$, so whatever behavior k ($[c']$, cs') has is what behavior $\text{match } (\text{Const } c) \text{ } cs \text{ } k$ should have. **Spec Satisfied!**

Cor. 16

For any almost total k and any value $c : \text{Sigma}$, $(\text{fn } cs \Rightarrow \text{match } (\text{Const } c) \text{ } cs \text{ } k)$ is almost total.

5 Plus Correctness

Inductive Assumption (Section 5)

For some $R1, R2 : \text{Sigma regexp}$,

$$\text{match } R1 \text{ } cs \text{ } k \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs \text{ such that } p \in \mathcal{L}(R1) \text{ and } k \text{ accepts } (p, s) \text{ with result } v. \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

$$\text{match } R2 \text{ } cs \text{ } k \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs \text{ such that } p \in \mathcal{L}(R2) \text{ and } k \text{ accepts } (p, s) \text{ with result } v. \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

for all values $cs : \text{Sigma list}$ and all almost total k .

In particular: if k is almost total, then so too are $(\text{fn } cs \Rightarrow \text{match } R1 \text{ } cs \text{ } k)$ and $(\text{fn } cs \Rightarrow \text{match } R2 \text{ } cs \text{ } k)$.

Lemma 17

For any almost total k , any $cs : \text{Sigma list}$,

$$\text{match (Plus(R1,R2)) cs k} \cong \begin{cases} v & \text{where (p,s) is a splitting of cs such that } p \in \mathcal{L}(\text{Plus(R1,R2)}) \text{ and } k \text{ accepts (p,s) with result } v. \\ \text{raise NoMatch} & \text{if there is no such (p,s)} \end{cases}$$

Cor. 18

For any almost total k , $(\text{fn cs} \Rightarrow \text{match (Plus(R1,R2)) cs k})$ is almost total.

6 Times Correctness

Inductive Assumption (Section 6)

For some $R1, R2 : \text{Sigma regexp}$,

$$\text{match R1 cs k} \cong \begin{cases} v & \text{where (p,s) is a splitting of cs such that } p \in \mathcal{L}(R1) \text{ and } k \text{ accepts (p,s) with result } v. \\ \text{raise NoMatch} & \text{if there is no such (p,s)} \end{cases}$$

$$\text{match R2 cs k} \cong \begin{cases} v & \text{where (p,s) is a splitting of cs such that } p \in \mathcal{L}(R2) \text{ and } k \text{ accepts (p,s) with result } v. \\ \text{raise NoMatch} & \text{if there is no such (p,s)} \end{cases}$$

for all values $cs : \text{Sigma list}$ and all almost total k .

In particular: if k is almost total, then so too are $(\text{fn cs} \Rightarrow \text{match R1 cs k})$ and $(\text{fn cs} \Rightarrow \text{match R2 cs k})$.

Lemma 19

For any almost total k , any $cs : \text{Sigma list}$,

$$\text{match } (\text{Times}(R1, R2)) \text{ } cs \text{ } k \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs \text{ such that } p \in \mathcal{L}(\text{Times}(R1, R2)) \text{ and } k \text{ accepts } (p, s) \text{ with result } v. \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

Cor. 20

For any almost total k , $(\text{fn } cs \Rightarrow \text{match } (\text{Times}(R1, R2)) \text{ } cs \text{ } k)$ is almost total.

7 Star Correctness

Inductive Assumption (Section 7)

For some $R : \text{Sigma regexp}$ and some $cs : \text{Sigma list}$

$$\text{match } R \text{ } cs \text{ } g \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs \text{ such that } p \in \mathcal{L}(R) \text{ and } g \text{ accepts } (p, s) \text{ with result } v. \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

$$\text{match } (\text{Star}(R)) \text{ } cs' \text{ } g \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs' \text{ such that } p \in \mathcal{L}(\text{Star}(R)) \text{ and } g \text{ accepts } (p, s) \text{ with result } v. \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

for all values $cs' : \text{Sigma list}$ of length less than cs , and all almost total g .

In particular: if g is almost total, then so too is $(\text{fn } cs \Rightarrow \text{match } R \text{ } cs \text{ } g)$. And if g is almost total, $\text{match } (\text{Star}(R)) \text{ } cs' \text{ } g$ for cs' of length less than cs .

■

Lemma 21

For any almost total k ,

$$\text{match } (\text{Star}(R)) \text{ } cs \text{ } k \cong \begin{cases} v & \text{where } (p, s) \text{ is a splitting of } cs \text{ such that } p \in \mathcal{L}(\text{Star}(R)) \text{ and } k \text{ accepts } (p, s) \text{ with result } v \\ \text{raise NoMatch} & \text{if there is no such } (p, s) \end{cases}$$

Cor. 22

For any almost total k , $(\text{fn } cs \Rightarrow \text{match } (\text{Star}(R)) \text{ } cs \text{ } k)$ is almost total.

8 match and LL Correctness

9 Reduction, Representation, and Printing Specification

Value Spec

`represent : (''S -> string) -> ''S regexp -> string`

REQUIRES: `toStr` is total, R is non-Zero

ENSURES: `represent toStr R` converts the regular expression R into POSIX-like regular expression syntax, using `toStr` to convert `Sigma` values into strings. This uses the convention of representing `One` as `(. {0,0})`

Defn. 23

A value $R : \text{Sigma } \text{regexp}$ is said to be **Zero-reduced** if either:

- $R = \text{Zero}$
- R does not contain any `Zeros`

Value Spec

`reduce : ''S regexp -> ''s regexp`

ENSURES: $(\text{reduce } R) \implies R'$ such that $\mathcal{L}(R) = \mathcal{L}(R')$ and R' is Zero-reduced (and also R' may have some superfluous `Ones` removed)

Value Spec

```
printRep : (''S -> string) -> ''S regexp -> unit
```

REQUIRES: toStr is total, $(\text{reduce } R) \not\Rightarrow \text{Zero}$

ENSURES: $\text{printRep toStr } R \Rightarrow ()$

EFFECT: Prints the value of $(\text{represent toStr } (\text{reduce } R))$ into the REPL.